Скрипты для Virgo. 

1.1. Зачем нам нужны скрипты?
Программа Virgo устроена таким образом, что может работать с любым приложением, удовлетворяющим стандартам Windows, без написания специальных скриптов. Однако многие программисты не обращают внимания на эти стандарты и создают свои собственные правила написания программ. Они программируют, например, кнопку, используя элемент «окно». Для зрячих людей это окно выглядит на экране как кнопка, поэтому их не волнует, какой это элемент.  Однако на брайлевский дисплей и синтезатор речи выводится элемент «окно», и слепой пользователь думает, что это окно, поэтому ему и в голову не придет нажать на него, как на кнопку. В этом случае скрипт изменяет имя элемента «окно» на имя «кнопка», и тогда слепой пользователь знает, что с этим делать. 
Кроме того, иногда необходимо собрать вместе несколько фрагментов информации, расположенных в различных частях экрана, чтобы показать их в компактной форме на брайлевском дисплее или воспроизвести в виде речи. Это тоже можно сделать с помощью скрипта.
Скрипты позволяют также получить доступ к внутренним функциям многих стандартных приложений, таких, например, как Word или Excel. 
Имеется различие между скриптами Windows и скриптами Virgo.
В то время как скрипты Windows используются для создания элементов Windows на экране, например, диалоговых окон, скрипты Virgo  нужны для влияния на существующие элементы на экране (например, для изменения их имен или порядка отображения на брайлевском дисплее). Это необходимо для того, чтобы объяснить слепому пользователю программные ситуации или правильно показать элементы, находящиеся на экране. Примером этого может служить диалоговое окно в Word 2000, которое появляется, когда Вы открываете файл. Курсор стоит в строке редактирования имени файла. Имя строки «Имя файла», однако, оно написано выше строки редактирования, в результате чего слепой пользователь нашел бы только пустую строку с курсором без понимания значения этой строки. Поэтому скрипт изменяет представление, и оба поля отображаются на брайлевской строке в горизонтальном порядке: сначала «Имя файла», затем пустое пространство для ввода имени файла.
В этой документации мы дадим Вам краткое введение в создание скриптов, чтобы Вы смогли самостоятельно делать небольшие адаптации прикладных программ. 
1.2.  Какие скрипты используются в Virgo? 

Где найти скрипты и под какими именами они хранятся? 
Все скрипты хранятся в папке «Scripts», которая находится в текущей папке Virgo (например, c:\Program Files\Virgo40\scripts).

Каждый скрипт должен иметь имя, совпадающее с именем приложения, для которого создан

этот скрипт, и расширение VBS. 
Таким образом, загружаемое приложение автоматически распознает скрипт и выполняет его.
Для того чтобы посмотреть или отредактировать скрипт  для какого-либо приложения, нужно загрузить это приложение и набрать на цифровом блоке команду N0/2 (уровень «слэш», команда «2»).  Нужный скрипт будет загружен в текстовый редактор, используемый в Virgo для скриптов.

1.3. Какой редактор использовать?
Как стандарт Virgo использует редактор Notepad.exe, встроенный в Windows. Однако может быть использован любой другой редактор, который запускается вместе с именем файла. Большинство редакторов обычно удовлетворяют этому условию. Мы рекомендуем редактор, в котором Вы можете переходить на строки  или столбцы непосредственно или где Вы, по крайней мере, можете получить информацию о точной позиции курсора.
Как назначить какой-либо редактор редактором скриптов Virgo?
1.
 Открыть файл phoenix.ini, который Вы найдёте в текущей папке Virgo. 
2.
 Добавить в разделе " [SYSTEM] " строку, указывающую путь к файлу редактора, например, "editor=c:\pfe\pfe32".
3.
 Сохранить файл, и перезагрузить Virgo.
С этого момента Virgo будет использовать данный редактор скриптов.

1.4. Краткое введение в Visual Basic
Для создания скриптов используется объектно-ориентированный язык программирования Visual Basic. В данном руководстве мы не ставим задачу обучить вас программированию на Visual Basic, поскольку этому посвящена специальная литература. Здесь мы дадим лишь краткий обзор некоторых, наиболее часто используемых в скриптах операторов Visual Basic, что позволит облегчить вам чтение существующих скриптов и написание своих собственных.

1.4.1. Оператор комментария
Оператор комментария не выполняет в скрипте никаких действий и может содержать произвольный текст. Этот оператор имеет две формы записи: самостоятельный оператор, начинающийся ключевым словом REM, за которым может следовать любой текст, или знак апострофа ('), который можно ставить после любого оператора. Примеры:

REM   Это комментарий

А=1  ' Это тоже комментарий

' Script Name    : DEFAULT.VBS
' Application    : DEFAULT-Script for Virgo

1.4.2. Оператор присваивания
 Общая форма оператора присваивания имеет вид:

ПЕРЕМЕННАЯ=ВЫРАЖЕНИЕ

Примеры:

DebugCnt = 0

DebugCnt = DebugCnt + 1

1.4.3. Оператор Dim

Используется для объявления переменной или массива переменных, которые можно будет использовать во всех процедурах скрипта. Внутри процедуры Dim используется для объявления локальной переменной, доступной только для этой процедуры.

Dim  ПЕРЕМЕННАЯ(РАЗМЕР), ПЕРЕМЕННАЯ(РАЗМЕР),…

Примеры:

DIM WName          ' Имя окна
DIM WLeft,WTop  ' Координаты окна
DIM WWidth           ' Ширина окна

DIM WHeight          ' Высота окна
1.4.4. Оператор Const
Используется для объявления константы, которую можно будет использовать во всех процедурах скрипта.
Const КОНСТАНТА=КОНСТАНТНОЕ ВЫРАЖЕНИЕ

Примеры:

Const DE = "1031"

Const RU = "1049"

Const xlEdgeLeft = 7

Const xlEdgeTop = 8

Const xlEdgeRight = 10

Const xlEdgeBottom = 9

Существуют также встроенные константы, например, True (Истина) и False (Ложь).

1.4.5. Оператор Set
Присваивает переменной ссылку на тот или иной объект.

Set ПЕРЕМЕННАЯ = ОБЪЕКТ

Примеры:

Set MOM = proc.GetMoFromFocus

Set ACol = ACh.ForeGroundColor
Set Char = ASE.GetAccCharacterFromIndex(1)
1.4.6. Условный оператор If
Используется для выполнения той или другой группы операторов в зависимости от значений условных выражений.

If  УСЛОВИЕ_1 Then                    ' Если УСЛОВИЕ_1 истинно, то

     БЛОК_ОПЕРАТОРОВ_1         ' выполняется БЛОК_ОПЕРАТОРОВ_1   

     ElseIf  УСЛОВИЕ_2 Then        ' Если УСЛОВИЕ_2 истинно, то

        БЛОК_ОПЕРАТОРОВ_2      ' выполняется БЛОК_ОПЕРАТОРОВ_2   
…..

Else                                                  ' иначе, если все условия ложны 
   БЛОК_ОПЕРАТОРОВ_ПО_УМОЛЧАНИЮ  ' выполняется 

                                                                                 ' БЛОК_ОПЕРАТОРОВ_ПО_УМОЛЧАНИЮ  

End If

Пример:

      If FindTopLevelWin("NameListWndClass") = 0 Then

          NameListActive = FALSE

          proc.Lst_Key = "+Down"

          proc.AddToCmdQueue("GOFOCUS") 

      Else

          E.SyncHere = FALSE

      End If

1.4.7. Оператор выбора Select Case
Этот оператор используется вместо последовательности операторов If.

Select Case ВЫРАЖЕНИЕ      ' Вычисляется значение ВЫРАЖЕНИЯ

       Case УСЛОВИЕ_1             ' Если вычисленное значение удовлетворяет данному условию,

                 БЛОК_ОПЕРАТОРОВ_1 ' то выполняется данный блок операторов и т.д.

       Case УСЛОВИЕ_2

                 БЛОК_ОПЕРАТОРОВ_2

       …….

       Case УСЛОВИЕ_n
                 БЛОК_ОПЕРАТОРОВ_n

       Case Else                                               ' В противном случае выполняется 

                 БЛОК_ОПЕРАТОРОВ_ПО_УМОЛЧАНИЮ  ' данный блок операторов

End Select
Запись Case УСЛОВИЯ может иметь одну из трех форм:

1) Case КОНСТАНТА_1, КОНСТАНТА_2, КОНСТАНТА_3…  

2) Case Is ЗНАК_ОТНОШЕНИЯ КОНСТАНТА                     
3) Case КОНСТАНТА_1 To КОНСТАНТА_2
Пример:

Select Case L    

       Case 3,5

                 Proc.SpeechOut "Три или пять"

                 Proc.BrailleOut "Три или пять"

       Case Is > 5

                 Proc.SpeechOut "Больше пяти"

                 Proc.BrailleOut "Больше пяти"

      Case –5 To 2

                 Proc.SpeechOut "От минус пяти до двух"

                 Proc.BrailleOut "От минус пяти до двух"

       Case Else      

                 Proc.SpeechOut "Ни одно из условий не выполнено"

                 Proc.BrailleOut "Ни одно из условий не выполнено"

End Select
Если ни одно из условий не выполнено, а ветви Case Else нет, выполнение программы может привести к непредсказуемым последствиям. Если значения ВЫРАЖЕНИЯ удовлетворяет сразу нескольким Case УСЛОВИЯМ, будет выполнено то из них, которое стоит ближе к началу оператора Select Case.

1.4.8. Оператор цикла For…Next
Этот оператор позволяет выполнить БЛОК_ОПЕРАТОРОВ определенное число раз.

For СЧЕТЧИК=X To Y Step Z

       БЛОК_ОПЕРАТОРОВ
Next СЧЕТЧИК
БЛОК_ОПЕРАТОРОВ будет выполнен столько раз, сколько значение шага Z нужно прибавить к начальному значению счетчика X, чтобы получить Y. Если значение шага не задано, то по умолчанию оно будет равно +1.

For I = 1 to AseColl.AseCount

         Set Ase = AseColl.GetASEFromIndex(I)

         Select Case Ase.Type

           Case 1

             Anz_Texte = Anz_Texte + 1

           Case 3

             Anz_Bitmap = Anz_Bitmap + 1

           Case 4

             Anz_Brush = Anz_Brush + 1

         End Select

      Next

1.4.9. Оператор цикла Do…Loop
Этот оператор позволяет повторять выполнение БЛОКА_ОПЕРАТОРОВ до тех пор, пока УСЛОВИЕ является истинным.

Do While УСЛОВИЕ

       БЛОК_ОПЕРАТОРОВ

Loop
или

Do 

       БЛОК_ОПЕРАТОРОВ

Loop While УСЛОВИЕ
или

Do Until УСЛОВИЕ

       БЛОК_ОПЕРАТОРОВ

Loop
или

Do 

       БЛОК_ОПЕРАТОРОВ

Loop Until УСЛОВИЕ

Предложение While (Пока) противоположно предложению Until (До). При использовании While выполнение БЛОКА_ОПЕРАТОРОВ цикла продолжается до тех пор, пока УСЛОВИЕ истинно (имеет значение True). 

При использовании Until выполнение БЛОКА_ОПЕРАТОРОВ цикла прекращается, как только УСЛОВИЕ принимает значение True. Для выхода из цикла прежде, чем будет достигнуто предложение Loop, можно поместить одну или несколько инструкций Exit Do внутри тела цикла, включив их в такие управляющие структуры, как If…Then…Else .

Примеры:

Do While HWnd <> 0

               CP = CP & System.GetClassName(HWnd) & "/"

               HWnd =System.GetParent(HWnd)

Loop

Do While ChildHWnd <> 0

      ClassName = System.GetClassName(ChildHWnd)

      If ClassName = "msctls_statusbar32" Then 

        Set WindowRect = System.GetWindowRect(ChildHWnd)

        S = Proc.GetString(WindowRect.Left, WindowRect.Top, _ 

            WindowRect.Right, WindowRect.Bottom)

        Exit Do 

      End If

      ChildHWnd = System.GetWindow(ChildHWnd, GW_HWNDNEXT) 

  Loop
1.4.10. Оператор цикла With
Этот оператор позволяет выполнить БЛОК_ОПЕРАТОРОВ, не повторяя имя указанного объекта в ссылках на объекты, свойства и методы. 

With ОБЪЕКТ

       БЛОК_ОПЕРАТОРОВ

End With

Примеры:

With System.GetAppObject(3).ActiveCell.Font

        If .Bold Then

          A = LngCon("Bold")

        End If

        If .Italic Then

          A = Trim(A) & " " & LngCon("Italic") & " "

        End If

 End With

With ASE.VisibleLocation 

      GriDL = .Left

      GridT = .Top

      GridR = .Right

      GridB = .Bottom

End With
1.4.11. Оператор обработки ошибок On Error
Этот оператор используется для включения режима перехвата ошибок, передачи управления в блок обработки ошибок или для того, чтобы игнорировать некоторые ошибки или совсем отключить режим перехвата ошибок.

On Error GoTo ИДЕНТИФИКАТОР_СТРОКИ

Используется для передачи управления в блок обработки ошибок, который вы можете создать в скрипте. ИДЕНТИФИКАТОР_СТРОКИ может быть либо номером строки, либо меткой.
On Error Resume Next

Используется для перехвата ошибок без передачи управления в блок обработки ошибок. Выполнение скрипта продолжается с оператора, непосредственно следующего за тем, который вызвал ошибку.

On Error GoTo 0

Используется для отключения режима перехвата ошибок.

1.4.12. Подпрограммы и функции

При написании скриптов используются два типа процедур – подпрограммы (процедуры Sub) и функции (процедуры Function). Процедуры обоих типов могут иметь аргументы – переменные, с помощью которых передаются значения, определяющие работу процедуры при конкретном вызове. Функция может возвратить единственное значение, а подпрограмма значений не возвращает.

Подпрограмма записывается следующим образом:

Sub ИМЯ_ПОДПРОГРАММЫ (АРГУМЕНТЫ)

     ПОСЛЕДОВАТЕЛЬНОСТЬ_ОПЕРАТОРОВ_1

     Exit Sub

     ПОСЛЕДОВАТЕЛЬНОСТЬ_ОПЕРАТОРОВ_2

End Sub
Функция записывается следующим образом:

Function ИМЯ_ФУНКЦИИ (АРГУМЕНТЫ)

     ПОСЛЕДОВАТЕЛЬНОСТЬ_ОПЕРАТОРОВ_1

     Exit Function
     ПОСЛЕДОВАТЕЛЬНОСТЬ_ОПЕРАТОРОВ_2

     ИМЯ_ФУНКЦИИ=ЗНАЧЕНИЕ
End Function

Наличие ключевого слова Public в описании подпрограммы или функции, указывает на доступность данной подпрограммы или функции для всех процедур скрипта.  Примеры подпрограмм и функций, используемых в скриптах, будут приведены ниже.   

1.5. Мой первый скрипт

Задача:
При нажатии команды N061 в программе Wordpad, текущее время должно быть отображено на брайлевской  строке и произнесено с помощью синтезатора речи.
Необходимо выполнить следующие шаги:
1. Запустите программу Wordpad.
2. Наберите на цифровом блоке команду N0/2 (уровень «слэш», команда «2»), чтобы загрузить скрипт для текущей программы (Wordpad) в стандартном редакторе скриптов. Если для этой программы не существует скрипта, Virgo загружает стандартный скрипт generic.vbs и переименовывает его в Wordpad.vbs.
3. Теперь добавьте в скрипт следующую функцию:
Public Function OnCmd (CMD)  

        If CMD = "L6N1" Then    ' Если набрана команда N061

          CMD = ""        ' Очистка 

          Proc.BrailleOut Time   ' Вывод текущего времени на брайлевскую строку

          Proc.SpeechOut  Time  ' Вывод текущего времени на синтезатор речи

       End If 

    End Function

4. Сохраните изменения.
Вернитесь с помощью Alt-Tab в Wordpad и нажмите N061.
Поздравляем Вас с Вашим первым скриптом!
1.6. Скрипт default.vbs
В папке «Scripts» находится также особый скрипт default.vbs, используемый Virgo по умолчанию. Наличие этого скрипта обязательно, т.к. без него невозможна нормальная работа  Virgo. 
1.7. Важные Virgo-Windows-компоненты
1.7.1. SDDP
Screen drawing data pump (sddp)
Этот компонент дает возможность Virgo наблюдать за экранными операциями.
SDDP связывается с системой на уровне драйвера графической карты. Он получает информацию от графического драйвера и сохраняет её в виде записей. 
SDDP работает как так называемый цепной драйвер, что означает, что он заставляет операционную систему поверить, что SDDP - драйвер графической карты. Этот метод, конечно, требует, чтобы установка SDDP проводилась после установки драйвера графической карты. Когда другой графический драйвер установлен после SDDP (например, потому что в компьютер должна быть установлена новая графическая карта), цепочка потеряется. Следовательно, SDDP должен быть установлен повторно. Однако это происходит автоматически, поскольку Virgo после перезагрузки обнаруживает новые драйверы и переустанавливает SDDP.
1.7.2. SED
Screen element data base (sed) – база данных элементов экрана
SED создает динамическую базу данных с информацией об объектах на экране, таких как растровые файлы, текстовые строки или рисунки.
Однако эти биты информации не имеют никаких контекстных связей. SED также известна как Offscreen Model  (закадровая  модель).
1.7.3. MOM
Meaningful object model (MOM) – модель значения объекта
MOM также создаёт динамическую базу данных, которая содержит и контекстную информацию. MOM обеспечивает объекты свойствами и методами, и таким образом может сравниваться с MSAA (см. 1.5.5).
MOM, например, содержит информацию о таких элементах Windows, как флажки, строки редактирования или кнопки.
1.7.4. SAPI
Speech application programming interface  (sapi) – интерфейс программирования речевых приложений
SAPI – это  интерфейс для управления синтезаторами. С помощью этого стандартного интерфейса Virgo может использовать любой речевой синтезатор, имеющий драйвер SAPI.
Основной элемент SAPI речи - так называемый движок. Движок по существу можно расценивать как речевой синтезатор. В движке происходит преобразование текста в речь (TTS). Движок  обеспечивает одни или несколько режимов вывода речевой информации.
Режим можно рассматривать как основной голос. 
Режим имеет параметры, такие как частота, основной тон, ударение и т.д., которые можно изменять. В итоге из режима получается голос, и это значит, что голос – это режим, адаптированный с помощью параметров к персональным потребностям пользователя.
SAPI поддерживает параллельное использование нескольких синтезаторов речи. Таким образом, Вы можете, например,  устанавливать три синтезатора, первый, потому что Вам больше всего нравится его движок для русского языка, второй, потому что он говорит на превосходном американском английском языке и третий, потому что у него такое естественное французское звучание. Два из них могут быть пакетами программ, в то время как третий может быть некоторым аппаратным средством,  подключённым к последовательному интерфейсу. В результате Вы будете иметь, по крайней мере, три движка, из которых Вы можете взять несколько режимов, а из них несколько голосов. Virgo позволяет Вам связывать различные ситуации с разными голосами и, таким образом Вы можете создать ваш собственный индивидуальный голосовой оркестр.
1.7.5. MSAA
MicroSoft Active Accessibility (MSAA) активный доступ Microsoft 
Первым шагом Microsoft в создании собственной модели доступа был MSAA. 
MSAA - технология, которая требует, чтобы прикладная программа поддерживала экранный доступ, для обеспечения оптимальной эффективности. MSAA работает по принципу клиент / сервер. 
Virgo - клиент, который требует необходимую информацию от сервера. Выполняемая прикладная программа - сервер. MSAA обеспечивает Virgo требуемой информацией об экране. Virgo и прикладная программа общаются друг с другом через интерфейс, называемый «Iaccessible».
MSAA представляет элементы интерфейса пользователя (например, таковые текстового процессора) как объекты Com. Эти объекты содержат так называемые свойства, которые описывают имя объекта, его позицию на экране и другую информацию, необходимую Virgo. 
Кроме того, объект содержит некоторые функции, так называемые методы.
Такой клиент, как Virgo может активизировать эти методы заставить объект выполнить некоторое действие. MSAA обеспечивает также уже существующие пользовательские интерфейсы операционной системы. Среди них – такие элементы как меню, полосы прокрутки и элементы диалоговых окон. Поэтому прикладные программы, использующие эти интерфейсы, также доступны без собственного интерфейса «Iaccessible».
Поддержка MSAA сервером необходима только тогда, когда он использует собственные пользовательские интерфейсы. На первый взгляд эта концепция выглядит очень хорошо, но она имеет серьезные недостатки. Прикладная программа должна быть специально разработана с поддержкой MSAA. Когда разрабатывается новая программа - это не проблема, но все старые программы едва ли получат выгоду из поддержки MSAA.
1.7.6. GUI
Grafical user interface (Gui) - графический интерфейс пользователя 
В графическом интерфейсе пользователя (Gui) экранная информация перехватывается в специальных точках операционной системы, так называемых обработчиках прерываний, и готовится к выводу через речь или брайлевский дисплей.
Virgo строит базу данных, содержащую всю экранную информацию.
Эта база данных обычно упоминается как заэкранная  модель.
Части экранной информации – это, например, видимые объекты, такие как символы, представляющие собой точечные рисунки со всеми их свойствами: тип объекта, шрифт, размер шрифта, начертание, цвет шрифта и фона, позиция пикселя, позиция фокуса и т.д.
1.8. Команды уровня скриптов N0/

(уровень «слэш» на цифровом блоке клавиатуры)
Уровень / Клавиша 1 = Загружает default.vbs - скрипт, заданный по умолчанию в редактор 
Уровень / Клавиша 2 = Загружает текущий скрипт в редактор 
Уровень / Клавиша 3 = Транслирует все скрипты 
Уровень / Клавиша 4 = 
Уровень / Клавиша 5 = Загружает файл DbLst.txt – список отладки в редактор
Уровень / Клавиша 6 = 
Уровень / Клавиша 7 = Показывает имя текущего приложения 
Уровень / Клавиша 8 = Вызывает диалог "Virgo Type Lib"
Уровень / Клавиша 9 = Вызывает окно диалога «Обзор объектов» 
 Уровень / Клавиша / = Показывает имя класса  МО-элемента, находящегося в фокусе 
Уровень / Клавиша SHIFT-/ = Показывает MO-координаты и копируют их в буфер
Уровень / Клавиша CTRL-/ = Показывает относительные MO-координаты и копирует их в буфер 
Уровень / Клавиша * = 
Уровень / Клавиша - = Выключить вывод информации об отладке скриптов  на брайль и речь
Уровень / Клавиша + = Включить вывод информации об отладке скриптов  на брайль и речь
Уровень / Клавиша E = Показывает текущий  MO-элемент 
2.  Как получить информацию?
2.1. Информации об отладке через брайлевский дисплей и речевой вывод
Virgo предлагает удобную возможность получать информацию из списка отладки через брайлевский дисплей и синтезатор речи. 
Для той цели существуют две специальных команды Virgo на уровне скриптов N0/ цифрового блока 
N0 / + = включить вывод информации отладки
N0 / - = выключить вывод информации отладки
Если Вы включите вывод информации отладки, Вы окажетесь в списке отладки. При написании и отладке скриптов Вы можете добавлять строки в этот список отладки самостоятельно,  используя в функциях и подпрограммах скрипта следующие процедуры:   
Proc. AddDebug "Строка"
или
Proc. ADB "Строка",
где "Строка" – это текстовая информация, которую Вы хотите поместить в список отладки.

Как двигаться в списке отладки:
Клавиша «Home» = Переход к самой первой строке списка,
Клавиша «End»  = Переход к самой последней строке списка,
Клавиша «Курсор вверх»   = Вверх по списку,
Клавиша «Курсор вниз»    = Вниз по списку 
Эти нажатия клавиш не будут действовать в прикладной программе, но любые другие ключи будут действовать в текущем приложении.
2.2. Экран и дампы окна
Virgo обеспечивает удобную возможность сохранять снимки отдельного окна или целого экрана. Таким образом, Вы можете анализировать ситуацию на экране, чтобы определить лучший способ обработки, который Вы будете использовать в Вашем скрипте.
Так называемые дампы сохраняются в папке Dump, расположенной в папке Virgo.
Имеются два вида дампов:
- ASE-Dump (ASE = Accessible Screen Element) - Доступный Экранный Элемент
- Line-Dump

Как  ASE-Dump, так и Line-Dump, могут быть выполнены как для отдельного окна, так  и для целого экрана.
ASE- Dump сохраняется с расширением ASE, Line-Dump – с расширением TXT, а в качестве имен файлов используются возрастающие номера.
В качестве специального параметра Virgo сохраняет также графический снимок целого экрана, используя файл дампа с расширением MBP.
В случае возникновения проблем Вы можете послать этот файл группе, занимающейся скриптами.
Для создания файлов дампа используют следующие команды Virgo:
N0 / Ctrl+1 = Дамп текущего окна (ASES)
N0 / Ctrl+2 = Дамп текущего  окна (Lines)
N0 / Shift+1 = Дамп текущего экрана (ASES)
N0 / Shift+2 = Дамп текущего экрана (Lines)
Примечание: Если Вы часто работаете с дампами, Вы должны время от времени освобождать

 каталог дампа, поскольку  файлы BMD довольно большие, приблизительно 500 КБ.
3. События в скриптах
3.1. Что такое события?

Создание скриптов для Virgo основано на событиях.

Событие – это то, что происходит с объектом в реальном времени и вызывает те или иные ответные действия. Например, событием является получение или потеря фокуса  приложением, нажатие клавиши на клавиатуре компьютера или кнопки на брайлевском дисплее и т.д. При написании скрипта вы определяете действия, которые должны быть выполнены в ответ на конкретное событие. При работе в среде Windows происходит много событий, однако каждый скрипт обрабатывает события, которые относятся к конкретному приложению. Например,  Вы хотите, чтобы приложение реагировало определенным образом на нажатие клавиши “Delete”. Тогда в скрипте,  связанном с данным приложением, Вы должны определить операции, которые должны быть выполнены в ответ на данное нажатие. Если скрипт не связан с данным событием, он это событие игнорирует. 

Обработка событий в скриптах осуществляется с помощью подпрограмм и функций, имеющих синтаксис Visual Basic.

Не все события одинаково важны. Ниже приведены некоторые из наиболее важных событий, вместе с примерами подпрограмм и функций обработки этих событий.
3.2.
 Подробно о событиях
3.2.1. Общие события
3.2.1.1. OnScriptInitialize
OnScriptInitialize (ThisScript)

Событие активизируется при инициализации скрипта во время запуска Virgo и может использоваться для установки параметров, загрузки ini-файлов и т.д.
Public Function OnScriptInitialize (ThisScript)

  ScriptName = ThisScript.Name

  Proc.AddDebug ScriptName & " инициализирован"

End Function
3.2.1.2. OnAppGotFocus
OnAppGotFocus
Событие активизируется, когда приложение получает фокус, например, если Вы открываете приложение или возвращаетесь в него с помощью Alt+Tab.
Public Function OnAppGotFocus (FocusChangeEvent)

  Proc.AddDebug ScriptName & " получил фокус"

  Active_Script = True

End Function

3.2.1.3. OnAppLostFocus
OnAppLostFocus
Событие активизируется, когда приложение теряет фокус, например, если Вы закрываете приложение или выходите из него в другое приложение. Это событие противоположно событию OnAppGotFocus .
Public Function OnAppLostFocus (FocusChangeEvent)

  Proc.AddDebug ScriptName & " потерял фокус"

  Active_Script = False

End Function

3.2.1.4. OnScriptTerminate
OnScriptTerminate 

Событие активизируется при закрытии Virgo. Может использоваться, чтобы сохранить параметры настройки приложения.
Public Function OnScriptTerminate (ThisScript)

  ScriptName = ThisScript.Name

  Proc.AddDebug ScriptName & " завершен"

End Function

3.2.2.
 События ввода 
3.2.2.1.
 OnPosswlow
OnPosswlow (Long) 

Событие активизируется, когда на брайлевском дисплее нажата позиционная кнопка. 
Пример 1:
Если в редакторе Notepad.exe в меню «Файл/Открыть» в фокусе находится командная кнопка,

то позиционные кнопки не должны реагировать. Рассмотрим функцию, которая это 

реализует. Формальный параметр L передает номер позиционной кнопки. 

Если L = 0, ни одна позиционная кнопка не нажата.
Public Function OnPosswlow (L)
  If Proc.fgwin_mo.Name =“Открыть”  and
     Proc.Focus_MO.Classname = "Button" Then
     'Когда заголовок текущего окна = "Открыть" и 
     'текущий элемент в фокусе имеет имя класса " Кнопка " 
     L = 0
     ' Дезактивировать позиционные кнопки 
  End If 

End Function
Пример 2:
В диалоге поиска (F3) редактора Notepad.exe позиционные кнопки 
должны выполнять следующие функции:
Первая позиционная кнопка - моделирует клавишу табуляции
Последняя позиционная кнопка - моделирует Shift+Tab
Вторая или предпоследняя позиционная кнопка - моделируют клавишу ENTER.
Public Function OnPosswlow(L)

   If Proc.fgwin_mo.Name = "Найти" Then

    ' если заголовок окна = "Найти"
          Select case L
         ' Проверьте значение L 
            case 1

            ' Если L = 1

               prog.AddToCmdQueue "SIMKEYTAB"

               ' Пошлите Virgo команду "SIMKEYTAB"- нажать Tab
             case Proc.BrlLen

             ' если L = длине строки брайлевского дисплея
               Prog. AddToCmdQueue "SIMKEYSHIFTTAB" 
               ' Пошлите Virgo команду "SIMKEYSHIFTTAB" - нажать Shift+Tab
              case 2, (Proc.BrlLen - 1)

              ' если L = 2 или (длине строки брайлевского дисплея – 1) 
                Prog. AddToCmdQueue "SIMKEYRET"
                ' Пошлите Virgo команду "SIMKEYRET" – нажать ENTER
          End Select 

   End If 

End Function
3.2.2.2. OnCmd
OnCmd (Variant)

Событие активизируется при наборе команды на цифровом блоке. 

Рассмотрим пример. Предположим, мы хотим, чтобы в некотором приложении в ответ на набор команды N061 (команда «1» на уровне «6») прозвучал гудок. 

Для этого в скрипт этого приложения добавим функцию Public Function OnCmd (S), в которой нас будет интересовать значение формального параметра S равное "L6N1", где "L6N1" – принятая в скриптах форма записи команды  N061 (L и N должны быть заглавными).
Public Function OnCmd (S)
    If  S = "L6N1"  Then 
       S = " "
       ' обнуление строки 
       Proc.DoBeep
       '  система производит гудок.
    End If
End Function

3.2.2.3. OnKeyPressed
OnKeyPressed (String)

Событие активизируется при нажатии ключа на клавиатуре ПК.

Под ключом подразумеваются все клавиши, кроме алфавитно-цифровых, а также их комбинации. Это функциональные клавиши F1-F12, клавиши управления курсором, а также клавиши Shift, Alt, Ctrl, Esc, Delete, Backspace, Insert, Enter, Tab. Ключами будут и комбинации этих клавиш, например, Ctrl+Shift, Shift+F1 и т.д.  

Рассмотрим пример функции, которая будет выдавать информацию о том, какой ключ нажат, в двух видах: в виде речевого сообщения и в виде сообщения на брайлевской строке. 

Для этого воспользуемся следующими процедурами:

 Proc.SpeechOut  S – для речевого вывода информации,

 Proc.BrailleOut S - для вывода информации на брайлевскую строку.

S – это переменная типа String, либо непосредственно текстовая строка.

Наша функция выглядит следующим образом:

Public Function OnKeyPressed(S)

   Proc.SpeechOut "нажат ключ" & S

   Proc.BrailleOut S

   S=""

End Function
Если мы добавим эту функцию в скрипт какого-либо приложения, например, в notepad.vbs для «Блокнота», и затем в этом приложении нажмем, например, F12, то синтезатор речи произнесет: «Нажат ключ доллар F12», а на брайлевской строке появится текст: «$F12».

«$F12» - это внутренний код Virgo для ключа F12. Аналогично мы можем увидеть коды для всех остальных ключей. Зная код ключа, мы можем использовать его, чтобы заставить наше приложение выполнять какое-либо действие только при нажатии именно этого ключа. Рассмотрим это на следующем примере. Изменим нашу функцию таким образом, чтобы по-прежнему, при нажатии любого ключа, его внутренний код выводился на брайлевскую строку, а на синтезатор речи выводился только код ключа F12.  

Public Function OnKeyPressed(S)

   If S="$F12" Then

      Proc.SpeechOut "нажат ключ" & S

   End If

   Proc.BrailleOut S

   S=""

End Function
3.2.2.4. OnKey
OnKey(ISystemKey)

Событие активизируется при нажатии клавиши на клавиатуре ПК перед обработкой этого нажатия приложением.
Функция Public Function OnKey(ISystemKey) позволяет анализировать нажатия клавиш прежде, чем передать их в приложение. Это дает возможность Virgo управлять работой приложения в зависимости от нажатия тех или иных клавиш клавиатуры ПК.

Параметр IsystemKey является объектом Virgo. Свойства этого объекта можно увидеть в «Мастере объектов» Virgo, который вызывается командой N0/9 (уровень «слэш», команда «9»). В списке объектов, расположенном в левом окне «Мастера объектов», ищем объект «SystemKey», а в правом окне видим список всех свойств этого объекта. Возьмем свойство «VkCode», которое позволит анализировать код нажатой клавиши. Рассмотрим функцию, в которой будем, например, выводить этот код на брайлевскую строку.  
Public Function OnKey(K)

    Proc.BrailleOut K.VkCode

End Function
3.2.2.5. OnBrlDK
OnBrlDK (String) 
Событие активизируется при нажатии клавиши на брайлевском дисплее.
Рассмотрим пример функции, которая в ответ на нажатие дисплейной клавиши D5 будет выполнять следующие действия: перемещение курсора на одну строку вверх, сдвиг на одну позицию вправо, чтение.

Public Function  OnBrlDK (S)

   If  S = "D5" then
     S = " "

     Proc. AddToCmdQueue "SIMKEYUP"   ' перемещение курсора на одну строку вверх
     Proc. AddToCmdQueue "SIMKEYRIGHT"   ' сдвиг на одну позицию вправо

     Proc. AddToCmdQueue "READDN"     '  чтение
  End If
End Function

3.2.2.6. OnBrl
OnBrl (String)

Событие активизируется при нажатии  комбинации клавиш на брайлевском  дисплее. 

Рассмотрим пример. Предположим, мы хотим, чтобы в некотором приложении в ответ на набор комбинации клавиш D1,2,3 прозвучал гудок.  Для этого в скрипт этого приложения добавим функцию Public Function OnBrl (S), аналогичную функции Public Function OnCmd (S).

Public Function OnBrl (S)
    If  S = "D1D2D3" then 
       S = " "
       ' обнуление строки 
       Proc.DoBeep
       '  система производит гудок.
    End If
End Function

3.2.2.7. OnBrlSW
OnBrlSW (String)
Событие активизируется при переключении переключателя на брайлевском дисплее.
3.2.2.8. OnPosSwitch
OnPosSwitch (SCR_PreClick_Class)

Событие активизируется при нажатии позиционной кнопки на брайлевском дисплее.
SCR_PreClick_Class является объектом Virgo. Свойства этого объекта можно увидеть в «Мастере объектов» Virgo.
3.2.2.9. OnPosOSLow
OnPosOSLow (VARIANT)
Событие активизируется при касании оптического сенсора на брайлевском дисплее перед выполнением функции, назначенной этому сенсору.
3.2.2.10. OnPosSensor
OnPosSensor (SCR_PreClick_Class)

Событие активизируется при касании оптического сенсора на брайлевском дисплее.
SCR_PreClick_Class является объектом Virgo. Свойства этого объекта можно увидеть в «Мастере объектов» Virgo.
3.2.2.11. OnMouseMove
OnMouseMove (PHX_Point_Class)

Событие активизируется при перемещении указателя мыши.
PHX_Point_Class является объектом Virgo. Свойства этого объекта, а также функции и подпрограммы для работы с этим объектом можно увидеть в «Мастере объектов» Virgo.
Для того, чтобы иметь возможность использовать обработку этого события в скрипте для какого-либо приложения, необходимо в Public Function OnAppGotFocus (FocusChangeEvent) скрипта данного приложения включить выражение:

Proc.MouseEventsEnabled = true

В Public Function OnAppLostFocus (FocusChangeEvent) нужно включить выражение:

Proc.MouseEventsEnabled = false, чтобы при выходе из приложения запретить обработку события OnMouseMove.
Ниже приведен пример функции для отображения на брайлевской строке текущих координат указателя мыши.

Public Function OnMouseMove(M)

    Proc.BrailleOut M.X & " / " & M.Y 

End Function
3.2.3. События вывода на брайль 

3.2.3.1.  OnStatusCell
OnStatusCell (VARIANT)

Событие активизируется при выводе информации в поле состояния.
3.2.3.2.  OnScrollEndLine
OnScrollEndLine (Boolean)
Событие активизируется, когда при навигации вправо достигается конец строки. 

3.2.3.3.  OnScrollBeginLine
OnScrollBeginLine (Boolean)
Событие активизируется, когда при навигации влево достигается начало строки.
3.2.4. События речевого вывода 
3.2.4.1. OnSpeechOutput 
OnSpeechOutput (SCR_PreOutput_Class)

Событие активизируется перед выводом речевого сигнала на громкоговоритель.
Свойства объекта SCR_PreOutput_Class можно увидеть в «Мастере объектов» Virgo.

Ниже приведен пример функции, которая выводит на брайлевскую строку свойство «Voice», являющееся обозначением номера голоса для Virgo (v1 – первый голос, v2 – второй голос и т.д.)

Public Function OnSpeechOutput (M)

    Proc.BrailleOut M.Voice 

End Function
3.2.4.2. OnSpeechDone
OnSpeechDone (LONG)

Событие активизируется после завершения вывода речевого сигнала на громкоговоритель.
3.2.5. События распознавания элементов

3.2.5.1. OnClass
OnClass (PHX_Object_Class)

Событие активизируется при переходе от одного элемента Virgo к другому.

Свойства объекта PHX_Object_Class можно увидеть в «Мастере объектов» Virgo. Воспользуемся двумя свойствами: U_M_ClsN и U_M_Role, чтобы показать, каким образом можно заставить Virgo изменить название элемента в каком-либо приложении. Предположим, мы хотим, чтобы вместо кнопки выбора было показано поле со списком, а при входе в меню раздавался системный гудок. Для этого добавим в скрипт приложения следующую функцию: 

Public Function OnClass (E)

  If  E.U_M_Clsn =  "#32768" Then    'Если это меню
    Proc.DoBeep                                    'Гудок
 End If

If  E.U_M_Role =  "OPT" Then            'Если это кнопка выбора
    E.U_M_Role = "CBO"                       'Присвоить название поля со списком 
End If

       End Function
3.2.5.2. OnControlName
OnControlName (SCR_ControlName_Class)
Событие активизируется при получении имени элемента. 

Функция  Function OnControlName может использоваться, например, для создания или изменения имени строки редактирования. 
Public Function OnControlName (CN)

 Dim MO

    Set MO=CN.PHX_O.EventMO 

  If CN.Role = "EDT" and MO.GetParent.ClassName = "rctrl_renwnd32" Then

   CN.d = Trim(Proc.GetBRLTextLine(0, (Proc.FGWinLeft), (CN.c_Top), (CN.c_Left) ,(CN.C_Top +20))) 

  End If 

End Function

3.2.6. События синхронизации

3.2.6.1. OnFocusMO
OnFocusMO (PHX_Object_Class)

Событие активизируется при смене фокуса МО.

Свойства объекта PHX_Object_Class можно увидеть в «Мастере объектов» Virgo.
Рассмотрим подпрограмму, которая в список отладки скриптов будет записывать такие свойства, как  U_M_ClsN – имя класса, и U_M_Role – название элемента.

Public Sub OnFocusMo(E)

 Proc.AddDebug " ClsN:" & E.U_M_ClsN & " Role: " & E.U_M_Role 

End Sub
Тогда в списке отладке мы найдем следующие строки:

ClsN: #32768 Role: POP   - для меню   

ClsN: Edit Role: EDT        - для строки редактирования

ClsN: ComboBox Role: CBO  - для поля со списком

ClsN: Button Role: OPT     - для кнопки выбора

ClsN: Button Role: BTN  - для кнопки действия
3.2.6.2. OnCursor
OnCursor (PHX_Object_Class)

Событие активизируется при смене позиции курсора.
3.2.6.3. OnFocusR
OnFocusR (PHX_Object_Class)

Событие активизируется при смене фокуса.
Рассмотрим пример подпрограммы изменения синхронизации.

Public Sub OnFocusR(E)

  'здесь мы проверяем, открылся ли информационный список (окно-VBA)
  Select Case E.U_M_Role
   Case "LI" 

    if  proc.FocusClassName = "VbaWindow" Then

    'это тот список, в который мы хотим синхронизировать
      NameListActive = TRUE 

      'установка флага, показывающего, что мы изменили синхронизацию
      E.SyncAs "LI"  'синхронизировать в качестве списка
    End If

   End Select

End Sub

3.2.6.4. OnHighLight
OnHighLight (PHX_Object_Class)

Событие активизируется при появлении новой подсветки.

Public Sub OnHighLight(E)

  Select Case E.U_M_Role 

    Case "WND"

      Select Case E.U_M_ClsN 

        Case "_sdm_XL8"

          E.U_M_Role   = "LI"

          E.U_M_Top    = E.Top

          E.U_M_Bottom = E.Bottom

          E.UseAs "M"

        Case "LI"

          E.U_M_Top    = E.Top

          E.U_M_Bottom = E.Bottom

          E.UseAs "M"

      End Select

  End Select

End Sub

3.2.6.5. OnControlSync
OnControlSync (PHX_Object_Class)

Событие активизируется при синхронизации на новом управляющем элементе.

Ниже приводится пример подпрограммы, обрабатывающей это событие.

Public Sub OnControlSync(E)

  If Proc.GalileoLoaded Then 

    If E.U_M_ClsN = "#32771" Then    ' Открыт диалог переключения задач (Alt-Tab)

      If LastOCS is nothing Then

        MouseToSet=4

        MouseToSetX=E.U_M_Left

        MouseToSetY=E.U_M_Bottom

        Set LastOCS=E

        Exit Sub

      End If

      If LastOCS.U_M_ClsN<>"#32771" Then     

        MouseToSet=4

        MouseToSetX=E.U_M_Left

        MouseToSetY=E.U_M_Bottom

        Set LastOCS=E

      End If 

    End If

    Set LastOCS=E

  End If
End Sub

3.2.6.6. OnNewFGWin
OnNewFGWin (PHX_Object_Class)

Событие активизируется при открытии нового окна.

Ниже приведен пример функции, которая заносит параметры нового окна в список отладки.

Public Function OnNewFGWin (E)

 ' Глобальным переменным присваиваются параметры окна
  WName = E.U_M_Name   ' имя окна
  WLeft = E.U_M_Left         ' расстояние от левого края экрана
  WTop = E.U_M_Top           ' расстояние от верхнего края экрана
  WWidth = (E.U_M_Right - E.U_M_Left)   ' ширина окна
  WHeight = (E.U_M_Bottom - E.U_M_Top)  ' высота окна
    If Set_Debug=1 then   ' если включен режим отладки 

      List.AddItem Lng_New_Win & " " & WName  'запись имени окна в список отладки
      List.ListIndex = List.ListCount - 1

      List.AddItem Lng_Pos & Lng_Left & WLeft & _ 'запись остальных параметров
      Lng_Top & WTop & _

      Lng_Width & WWidth & _

      Lng_Height & WHeight

      List.ListIndex = List.ListCount - 1

    End If
End Function

4. Virgo-объекты

Концепция объектного программирования позволяет работать с объектами. 

В Windows объектами являются такие элементы, как кнопки, окна, поля, меню и даже целые приложения. Понятие объекта включает в себя параметры объекта (свойства) и программные средства для работы с ним (методы). Одинаковые объекты могут объединяться в группы или классы. Свойство объекта записывается в виде:

ОБЪЕКТ.СВОЙСТВО

Например, если Win – это объект Virgo, являющийся окном, то такое свойство этого объекта, как имя, записывается в Virgo следующим образом: Win.U_M_Name 

Для создания, уничтожения или воздействия на объекты используются различные методы. 

Метод – это программа, которая выполняет определенные действия над объектом. Вызов метода, выполняющего действия над тем или иным объектом, записывается следующим образом:

ОБЪЕКТ.МЕТОД

Все объекты Virgo, вместе со свойствами, подпрограммами и функциями, связанными с этими объектами, можно найти в «Мастере объектов» Virgo, который вызывается командой N0/9 (уровень «слэш», команда «9»). 
4.1.
 Объект "System"
4.1.1.
 Sub SimulateMouse
Sub SimulateMouse(ByVal Action As Long, ByVal xPos As Variant, ByVal yPos As Variant)

system.SimulateMouse(Action, xPos, yPos)

Описание:
Выполняет щелчок мышью в точке с координатами xPos и yPos.
Параметры:
Action = может принимать значение одной из следующих констант (шестнадцатеричных):
const mouseMove
= 0x1;

const mouseLeftButtonDown = 0x2;

const mouseLeftButtonUp
= 0x4;

const mouseRightButtonDown = 0x8;

const mouseRightButtonUp = 0x10;

const mouseMiddleButtonDown = 0x20;

const mouseMiddleButtonUp = 0x40;

const mouseLeftButtonClick = 0x80;

const mouseRightButtonClick = 0x100;

const mouseLeftButtonDoubleClick = 0x200;

const mouseRightButtonDoubleClick = 0x400;

const mouseWeel = 0x800;

const mouseRelativeMove
= 0x1000;
Xpos - значение координаты X, на которой должен быть выполнен щелчок мышью.
YPos - значение координаты Y, на которой должен быть выполнен щелчок мышью.
Пример 1:
Нужно выполнить щелчок правой кнопкой мыши на кнопке «Пуск».
Public Sub ClickToStart()

  System.SimulateMouse 257, 2, 576

  '(mouseMove + mouseRightButtonClick), 2, 576

End Sub

Пример 2:
В редакторе Notepad.exe нужно выполнить щелчок левой кнопкой мыши на пункте меню «Файл», чтобы открыть меню «Файл».
В нашем первом примере мы произвели щелчок в абсолютных координатах. Это не является лучшим выбором, потому что когда окна перемещаются, координаты меняются  и щелчок мог бы быть выполнен в неправильном месте. 
Чтобы избежать этого, в данном примере мы используем относительные координаты.
public sub ClickToFile()

  System.SimulateMouse 129, proc.fgWinLeft + 4, proc.FgWinTop + 23

  '(mouseMove + mouseLeftButtonClick), FgWinLeft + 4, FgWinTop + 23

end sub

4.2. Объект "PHX_Proc_Class"
Свойства, подпрограммы и функции для этого объекта можно увидеть в «Мастере объектов». Многие из них часто используются в различных подпрограммах и функциях скриптов.

Рассмотрим примеры. 

4.2.1. Sub DoBeep 

Sub DoBeep ( )

Proc.DoBeep 

Производит гудок. Полезно использовать для отладки скриптов.
4.2.2. Property ForeGroundApp
Property ForeGroundApp As String
Proc.ForeGroundApp
Возвращает имя актуального приложения.
4.2.3. Sub BrailleOut
Sub BrailleOut (S As Variant)

Proc. BrailleOut "S"

Вывод информации "S" на брайлевскую строку.
4.2.4. Sub SpeechOut
Sub SpeechOut (S As Variant)

Proc.SpeechOut "S"

Вывод информации "S" на синтезатор речи.
4.2.5. Property HotkeysActive
Property HotkeysActive As Long
Proc.HotkeysActive
Цифровой блок клавиатуры включен.
4.2.6. Property HotkeyState
Property HotkeyState As Long
Proc.HotkeyState 
Сообщает состояние цифрового блока клавиатуры.
4.2.7. Property Key_LS_Down
Property Key_LS_Down As Boolean
Proc. Key_LS_Down 

Нажата клавиша «Левый Shift» (True или False).

4.2.8. Property Key_LC_Down
Property Key_LC_Down As Boolean
Proc.Key_LC_Down 

Нажата клавиша «Левый Ctrl» (True или False).

4.2.9. Property Key_LA_Down
Property Key_LA_Down As Boolean
Proc.Key_ LA _Down 

Нажата клавиша «Левый Alt» (True или False).
4.2.10. Property FgWinTitle
Property FgWinTitle As String
Proc.FgWinTitle    

Возвращает заголовок окна.
Пример: 
Proc.BrailleOut Proc.FgWinTitle     'Выдает заголовок окна на брайлевскую строку.
 4.2.11. Моделирование команд Virgo
Чтобы в скрипте была выполнена команда Virgo, нужно добавить в подпрограмму или функцию скрипта следующую строку:

 Proc.AddtoCmdQueue "VirgoCmd"

Где "VirgoCmd" – это команды Virgo, которые можно найти в файле Virgo40\lng\1049\cmdnames.ini или в меню «Назначение клавиш» диалога конфигурации Virgo.

Например,

Proc.AddToCmdQueue "SIMKEYUP"   ' курсор вверх

Proc.AddToCmdQueue "SIMKEYRIGHT"  ' курсор вправо

Proc.AddToCmdQueue "READDN"   ' начать непрерывное чтение 

5.
 Некоторые примеры скриптов
5.1.
 Общие замечания по примерам
Тестовая программа Bsp1.exe, реализующая примеры, приведенные ниже, находится в папке Bsp1. После стандартной установки Virgo эта папка должна быть в
 c:\Program Files\Virgo40\doc\ru - для русского языка, En - для английского языка, De – для немецкого языка и т.д.
В папке вместе с программой, реализующей примеры, Вы также найдете готовый скрипт для этой программы.
5.2. Пример 1
Тема: 

-  Назначить Controltype 
-  Синхронизировать события
-  Найти в ASE-совокупности некоторый цвет
Запустите программу BSP1.exe.
Двигайтесь с помощью клавиши табуляции по окну программы. Вы найдете кнопку  «Ende», нажав на которую Вы можете закрывать программу.
Помимо этого Вы также попадете на элемент Windows, который Virgo не может распознать.
Теперь щелкните один раз на «001 Zeile», и включите Мастер Экрана. С помощью Мастера Экрана Вы, конечно, выясните, что имеете здесь список, который не отслеживается программой Virgo.
Первая задача для нашего скрипта будет состоять в том, чтобы заставить Virgo отслеживать этот список. Поместите курсор перед «Z» в строке «001 Zeile» и введите F, чтобы отобразить цвет. Virgo отобразит цвет «0», а когда Вы переместите курсор на одну строку вниз, Virgo сообщит «FFFF». Следовательно, в скрипте мы должны будем найти цвет «0» и переместить курсор в эту точку. Для той цели мы используем событие OnSystemIdle.
Public Sub OnSystemIdle (E)

  Dim C
  ' Переменная совокупности
  E = True
  ' чтобы снова вызывать OnSystemIdle 
    Set C = Proc.Last_M_Event.GetASETextCollForMO
  ' Переменная С  связывается с ASE- совокупностью

   ShowASEs(С)
  ' Вызов самодельной подпрограммы, которая содержит ASE- совокупность как параметр 
End Sub
Public Sub ShowASEs(ASECol)

  Dim I, C, ASE

  For I = 1 TO ASECol.ASECount ' цикл по элементам  ASE- совокупности
      Set ASE = ASECol.GetASEFromIndex(I)

      C =  Proc.GetPixelColor(ASE.VisibleLocation.Left - 1, ASE.VisibleLocation.Top +1)

      ' Цвет около левого верхнего угла ASE
            If Hex(C) = "0" Then 

             ' если найден цвет фокуса
                  Proc.AddEvent "F", ASE.VisibleLocation.Left, ASE.VisibleLocation.Top +1, _ 

                   ASE.VisibleLocation.Right, ASE.VisibleLocation.Bottom - 1

                ' Моделирует синхронизацию фокуса с соответствующими координатами
            End If

    Next  

 End Sub

Введите эти подпрограммы в ваш скрипт и Вы увидите, что Virgo будет отслеживать список.
Однако Virgo не показывает нам, что это список.
Для той цели мы должны в подпрограмме OnClass определить, что тип элемента с именем класса «MSFlexGridWndClass» должн быть отображен как «LST» (список).
Но как мы можем узнать имя класса этого элемента?
Просто поместите фокус на список, и нажмите N0//. Вы увидите имя класса: «MSFlexGridWndClass».
Public Sub OnClass(E) 

   IF E.U_M_ClsN = "MSFlexGridWndClass" THEN

      ', если имя класса  = " MSFlexGridWndClass "
      E.U_M_Role = "LST"

      ' присвоить этому элементу тип "LST" (список) 
   END IF

End Sub

